/*
 * Copyright (C) 2011 Nokia Corporation and/or its subsidiary(-ies).
 * All rights reserved.
 *
 * This file is part of Harmattan SmashMiner OpenGL game application.
 * 
 * Harmattan SmashMiner OpenGL Example Application version 1.0.0
 *  
 * Latest update: 15.4.2011
 *
 * The Harmattan SmashMiner OpenGL example application demonstrates how to use 
 * the OpenGL ES in Harmattan devices.
 *
 * This example is provided as a starting point for 3rd party
 * developers (you) to ease the implementation of OpenGL based 
 * games. Ideas, parts of code or methodologies that this
 * example application uses can be freely used without restrictions.
 *
 * See file README.txt how to build and compile this example application
 * in the Harmattan SDK environment. 
 * 
 * See file INSTALL.txt to find out what is required to run this
 * application and how to install the application to the device or
 * alternatively in QEMU-emulator.
 *
 * The LICENSE.txt file explains the license this software is provided
 * with. A copy of the same license is included in this source file.
 * 
 */

/*
 * Copyright (C) 2011 by Nokia Corporation.
 *
 * Permission is hereby granted, free of charge, to any person obtaining a copy
 * of this software and associated documentation files (the "Software"), to deal
 * in the Software without restriction, including without limitation the rights
 * to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
 * copies of the Software, and to permit persons to whom the Software is
 * furnished to do so, subject to the following conditions:
 *
 * The above copyright notice and this permission notice shall be included in
 * all copies or substantial portions of the Software.
 *
 * THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
 * IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
 * FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
 * AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
 * LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
 * OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
 * THE SOFTWARE.
 */

import Qt 4.7

// Switch QML element that can be used to select boolean "On"-"Off" values.
Rectangle {
    id: buttonSwitch

    property bool on: false

    function toggle() {
        if (buttonSwitch.state == "on")
            buttonSwitch.state = "off";
        else
            buttonSwitch.state = "on";
    }

    function releaseSwitch() {
        if (knob.x == 1) {
            if (buttonSwitch.state == "off") return;
        }
        if (knob.x == (buttonSwitch.width-knob.width)) {
            if (buttonSwitch.state == "on") return;
        }
        toggle();
    }

    width: 60
    height: 25
    color: "#00000000"
    visible: true
    opacity: 100

    Image {
        id: background

        anchors.fill: parent
        source: "pics/background.svg"

        MouseArea {
            anchors.fill: parent;
            onClicked: toggle()
        }
    }

    Image {
        id: knob

        x: 1;
        y: 2
        width: parent.height
        height:  parent.height
        anchors.verticalCenter: parent.verticalCenter
        source: "pics/knob.svg"

        MouseArea {
            anchors.fill: parent
            drag { target: knob; axis: Drag.XAxis; minimumX: 1; maximumX: buttonSwitch.width-knob.width }
            onClicked: toggle()
            onReleased: releaseSwitch()
        }
    }

    Text {
        id: onText

        text: qsTr("On")
        visible: false
        color: "white"
        font.pointSize: 13
        anchors { left: background.right; leftMargin: 10; verticalCenter: parent.verticalCenter }
    }

    Text {
        id: offText

        text: qsTr("Off")
        visible: true
        color: "grey"
        font.pointSize: 13
        anchors { left: background.right; leftMargin: 10; verticalCenter: parent.verticalCenter }
    }

    states: [
        State {
            name: "on"
            PropertyChanges { target: knob; x: buttonSwitch.width-knob.width }
            PropertyChanges { target: buttonSwitch; on: true }
            PropertyChanges { target: onText; visible: true }
            PropertyChanges { target: offText; visible: false }
        },
        State {
            name: "off"
            PropertyChanges { target: knob; x: 1 }
            PropertyChanges { target: buttonSwitch; on: false }
            PropertyChanges { target: onText; visible: false }
            PropertyChanges { target: offText; visible: true }
        }
    ]

    transitions: Transition {
        NumberAnimation { properties: "x"; easing.type: Easing.InOutQuad; duration: 200 }
    }
}
